

**NASA
Technical
Paper
2396**

November 1984

NASA-TP-2396 19850004525

**Algorithm for Astronomical,
Extended Source, Signal-to-
Noise Ratio Calculations**

R. R. Jayroe

LIBRARY COPY

1984

LANGLEY RESEARCH CENTER
LIBRARY, NASA
HAMPTON, VIRGINIA

NASA

**NASA
Technical
Paper
2396**

1984

Algorithm for Astronomical,
Extended Source, Signal-to-
Noise Ratio Calculations

R. R. Jayroe

*George C. Marshall Space Flight Center
Marshall Space Flight Center, Alabama*



National Aeronautics
and Space Administration

Scientific and Technical
Information Branch

TABLE OF CONTENTS

	Page
GENERAL DESCRIPTION	1
USER INSTRUCTIONS	1
EQUATIONS USED	3
REFERENCES	7
APPENDIX A - OUTPUT EXAMPLE	10
APPENDIX B - PROGRAM LISTING	12

TECHNICAL PAPER

ALGORITHM FOR ASTRONOMICAL, EXTENDED SOURCE, SIGNAL-TO-NOISE-RATIO CALCULATIONS

GENERAL DESCRIPTION

The software computes signal-to-noise ratios (S/N) as a function of observation time and observation times as a function of signal-to-noise ratios for viewing an astronomical extended source with a telescope and focal plane detector array outside the Earth's atmosphere. The extended source is assumed to be square or rectangular, uniform in brightness, and centered on the optic axis of the telescope. A combined S/N is computed for all the detector picture elements (pels) affected by the extended source, as well as one for a single pel, which may be completely or only partially filled (at the user's discretion) by the extended source. The user is also allowed to shift the extended source image in any direction by an amount equal to or less than the pel dimensions.

The cosmic background within the field-of-view is also assumed to be uniform in brightness. The user may request a monochromatic or polychromatic extended source and background. The user also has the choice of requesting a uniform or blackbody spectral distribution for the source and background.

The optical system is characterized by the telescope entrance pupil diameter and the system focal ratio. The user can also include optical filters by specifying their transmission characteristics.

The focal plane detector is assumed to be a charge coupled device (CCD) array that can have square or rectangular shaped pels and dead spaces between the pels. The CCD is also characterized by a root mean square (rms) readout noise, dark current, and quantum efficiency.

The program has the option to change some of the parameters in the calculations without restarting and completely re-typing all of the input data.

USER INSTRUCTIONS

The user is first asked to input the visual magnitude of the extended source and then to choose whether or not the spectral distribution of the source is uniform or blackbody. The cosmic background will also have the same type spectral distribution. If a blackbody distribution is chosen, the user is asked to input the effective temperature ($^{\circ}\text{K}$) of the source. The effective temperature for the cosmic background will be requested later.

Next the user is asked to input the lower and upper wavelengths (cm) and the number of wavelengths to consider. For Space Telescope, the lower and upper wavelengths are 10^{-5} cm and 11×10^{-5} cm. The maximum number of wavelengths that the program can handle is 21, which makes the wavelength increment 0.5×10^{-5} cm.

For this case, the filter functions ($0 \leq \text{transmission} < 1$) and quantum efficiencies (electrons/photon) must be specified and read in (all 21 of them) in increments of 0.5×10^{-5} cm. The user is asked if there is a table of quantum efficiencies and filter functions. If the answer is yes, then two files of data must have been previously prepared, which are now read into the program. One file contains the quantum efficiencies and the other contains the filter transmissions. Because the program integrates over wavelength, the first and last non-zero filter transmissions should be divided by two in order to use the trapezoidal rule for numerical integration. If the answer is no, the user is assumed to desire a system with 100 percent quantum efficiency and no filters.

To obtain a monochromatic source and background, the lower wavelength will be chosen as the desired wavelength, if the number of wavelength calculations is set to one. The user is then asked to input the quantum efficiency and filter transmission for that wavelength.

Next, the user is asked to input the diameter (cm) of the telescope entrance pupil and the focal ratio of the optical system. Then the visual magnitude of the cosmic background is requested. If a blackbody distribution was requested, an effective background temperature ($^{\circ}\text{Kelvin}$) will also be requested.

The user is then asked to specify the characteristics of the CCD pel array and the location of the extended source image within the array. To visualize the effects of the following inputs, see Figure 1 and consider the definitions listed below:

XSW = X direction width of the extended source in arc sec

YSW = Y direction width of the extended source in arc sec

XPW = X direction width of the CCD pel in arc sec

YPW = Y direction width of the CCD pel in arc sec

XPD = X distance between pel centers in arc sec, $YPD \geq XPW$

YPD = Y distance between pel centers in arc sec, $YPD \geq YPW$

XOF = X distance offset in arc sec of extended source image from the origin,
 $0 \leq XOF \leq XSW$

YOF = Y distance offset in arc sec of extended source image from the origin,
 $0 \leq YOF \leq YSW$

Dead space between the pels is created whenever $XPD > XPW$ and/or $YPD > YPW$, i.e., the pel center separation is greater than the pel dimension in the same direction. The program computes the size of the array needed based upon the size and location of the extended source image. All parts of the source image fall either upon a dead space or a pel.

After entering the x,y pel widths, x,y source image widths, x,y pel center separations, and the x,y source image offsets, the program computes a combined signal-to-noise ratio for an area that contains all pels effected by the extended source image. In the case of Figure 1, a combined signal-to-noise ratio would be computed for a three by three array. The user can obtain a S/N for a single pel completely

filled by the portion of the source image by entering a "1," when the prompt "Enter Fraction of a Pel for S/N Per Pel" is received. To obtain the S/N for a pel partially filled by the portion of the extended source image, the user enters the fraction of the pel that is filled by the portion of the extended source. To complete the detector characteristics, the user inputs the rms readout noise in the electrons per pel and the detector temperature in °Kelvin. For the particular CCD array used with the Space Telescope and the Wide Field/Planetary Camera, the readout noise has been suggested [1] to range from 13.9 to 17.8 electrons per pel rms. For the above CCD array, the maximum operating temperature [1] is expected to be around 178°K (-95°C), and the dark current in electrons per pel per second is computed from this input temperature. The user can input a mean dark current directly by first entering a temperature less than 4°K. A prompt will then appear requesting the desired dark current.

The next set of inputs is concerned with the output data that is desired. The user first specifies the observation start and end times in seconds and the number of time calculations. For example, to compute the S/N in hour intervals from 1 to 14 hr, the user would input a start time of 3600 sec, and an end time of 50400 sec and 14 as the number of time calculations. To reverse the situation and determine the observation times needed to achieve a desired range of S/N, the user inputs a start and end S/N and the desired number of S/N calculations. The program will proceed to output the desired data.

After the output is complete, the user may repeat the calculations for a source with a different visual magnitude and/or a background with a different visual magnitude, without having to re-input all the previous input data. An additional repeat calculation that can be accomplished with or without the magnitude changes, is to change the characteristics of the detector. This option, however, requires that the user re-input all data starting with the detector characteristics and continuing through the number of S/N calculations. The program terminates when the user answers no to all the repeat calculation options.

EQUATIONS USED

The system of units used is the cgs system, except for the image source and pel dimensions, which are in arc-sec. The total energy emitted in the eye responsive spectral region per square arc-sec (\square) from an extended, astronomical source of visual magnitude m and received per square meter per second outside the Earth's atmosphere is given by [2]

$$I_m^* = 2.54 \times 10^{-6} \times 10^{-0.4m} \text{ lux} / \square \quad . \quad (1)$$

Equation (1) is converted from photometric to radiometric units by dividing by

$$0.68 \int_{3.8 \times 10^{-5} \text{ cm}}^{7.6 \times 10^{-5} \text{ cm}} K(\lambda) d\lambda \text{ lux per (ergs cm}^{-2} \text{ sec}^{-1} \Delta \lambda^{-1}) \quad , \quad (2)$$

where $\Delta\lambda$ is the wavelength interval (expressed in cm) and $K(\lambda)$ is the photopic eye response [2] given in Table 1. Dividing equation (1) by equation (2) is tantamount to assuming a uniform spectral distribution of the extended source and gives

$$I_m(\lambda) = \frac{3.7353 \times 10^{-6} \times 10^{-0.4 m} \text{ ergs cm}^{-2} \text{ m}^{-1} \text{ sec}^{-1} \Delta\lambda^{-1}}{\int_{3.8 \times 10^{-5} \text{ cm}}^{7.6 \times 10^{-5} \text{ cm}} K(\lambda) d\lambda} \quad (3)$$

If $B(\lambda, T)$ is the blackbody distribution having an effective temperature $T(^{\circ}\text{Kelvin})$, then equation (3) can be converted to a blackbody distribution in the following manner [3]:

$$I_m(\lambda, T) = \frac{3.7353 \times 10^{-6} \times 10^{-0.4 m} B(\lambda, T) \text{ ergs cm}^{-2} \text{ m}^{-1} \text{ sec}^{-1} \Delta\lambda^{-1}}{\int_{3.8 \times 10^{-5} \text{ cm}}^{7.6 \times 10^{-5} \text{ cm}} K(\lambda) B(\lambda, T) d\lambda} \quad (4)$$

Any type distributions can be used for $B(\lambda, T)$ in equation (4) since multiplying equation (4) by $0.68 K(\lambda)$ and integrating from $3.8 \times 10^{-5} \text{ cm}$ to $7.6 \times 10^{-5} \text{ cm}$ reproduces equation (1). If $B(\lambda, t)$ were a uniform distribution, then equation (3) would be obtained again. Assume that a uniform distribution has been selected and that the following quantities have been specified:

$d\Omega$ = area of the extended source in m^2

A = area of the telescope entrance pupil (cm^2)

hc/λ = ergs/photon

$Q(\lambda)$ = quantum efficiency in electrons per photon

$F(\lambda)$ = optical filter transmission .

The quantum efficiencies [1] used in the output example are shown in Table 2. They include the combined effects of the optical telescope assembly and the wide field and planetary camera. The filter function [4] used in the output example is listed in Table 3.

The rate at which electrons are liberated from the detector per wavelength interval at a chosen wavelength is given by

$$S_s(\lambda) = \frac{A(d\Omega)}{hc} \lambda I_m(\lambda) Q(\lambda) F(\lambda) \text{ electrons sec}^{-1} \Delta\lambda^{-1} \quad (5)$$

Equation (5) is called the monochromatic source signal current. Integrating equation (5) over the wavelength region influenced by the quantum efficiency and optical filter produces the polychromatic signal current. If there is a dead space between the detector pels, $d\Omega$ is reduced to account for the dead space and any image offsets input by the user. The polychromatic source signal current is given by

$$S_s = \frac{A(d\Omega)}{hc} \int_{\lambda_1}^{\lambda_2} I_m(\lambda) Q(\lambda) F(\lambda) \lambda d\lambda \text{ electrons sec}^{-1} \quad (6)$$

By the same analogy, the polychromatic, cosmic background, signal current is given by

$$S_b = \frac{A(N_p d\Omega')}{hc} \int_{\lambda_1}^{\lambda_2} \lambda I_m'(\lambda) Q(\lambda) F(\lambda) d\lambda \text{ electrons sec}^{-1} \quad (7)$$

where m' could be different visual magnitude than that of the source, $d\Omega'$ is the area of single pel in \square , and N_p is the number of pels that have been totally or partially illuminated by the extended source image. Thus, $(N_p d\Omega')$ is always equal to or greater than $(d\Omega)$.

To compute a signal-to-noise ratio, the contributions to the signal term and the noise term must be defined. For the polychromatic case, for example, the signal is equation (6) multiplied by the time in seconds. If the incoming photon flux is assumed to be Poisson distributed, then the variance of the signal is also equal to the mean signal. If the signal to noise ratio is defined as the ratio of the mean to the standard deviation of the mean, then the source signal will contribute to the noise. Other contributions to the noise include the cosmic background signal, the rms readout noise in electrons per pel and the mean dark current in electrons per pel second. If all of the noise contributors are assumed to be statistically independent, then the signal-to-noise ratio is given by

$$(S/N) = \frac{S_s \cdot t}{(S_s \cdot t + S_b \cdot t + N_p \cdot R^2 + N_p \cdot D \cdot t)^{1/2}} \quad (8)$$

where R is the rms readout noise, N_p is the total number of pels that are totally or partially illuminated by the extended source, D is the dark current and t is the time in seconds. As previously mentioned, the user has the option of inputting the dark current directly or inputting the detector temperature and having the dark current computed for the CCD array on Space Telescope for the Wide Field/Planetary Camera. The equations [5] for the dark current are given by

$$D = 29.3 \times 10^9 \times T^{3/2} \exp(-5802.1Z) , \quad (9)$$

where

$$Z = \frac{1.1557}{T} - \frac{7.021T \times 10^{-4}}{1108+T} , \quad (10)$$

where T is the detector temperature in °Kelvin.

Equation (8) is a combined signal-to-noise ratio for all pels totally or partially illuminated by the extended source. To compute the signal to noise ratio for one pel that is totally or partially illuminated by the source, equation (8) is rewritten as

$$(S/N)' = \frac{S_s' \cdot t \cdot f}{(S_s' \cdot t \cdot f + S_b \cdot t + R^2 + D \cdot t)^{1/2}} \quad (11)$$

where f is the fraction of a pel illuminated by the source, and S_s' is equation (6) except that $d\Omega$ has been replaced with $d\Omega'$ as in equation (7). Equations (8) and (11) give the signal-to-noise ratios as a function of observation time. However, the time can be solved for in terms of the signal to noise ratio. Solving equation (8) for time gives

$$t = (S/N)^2 \frac{(S_s + S_b + N_p \cdot D)}{2S_s^2} + \left\{ (S/N)^4 \frac{(S_s + S_b + N_p \cdot D)^2}{4S_s^4} + (S/N)^2 N_p (R/S_s)^2 \right\}^{1/2} \text{ sec} \quad (12)$$

Equation (11) can be similarly written by setting $N_p = 1$ and replacing S_s with $S_s' f$.

REFERENCES

1. Westphal, J. A.: The Wide Field/Planetary Camera. NASA CP-2244, August 1982.
2. Allen, C. W.: Astrophysical Quantities. The Athlone Press, Third Edition, 1973.
3. Ramsey, R. C.: Spectral Irradiance from Stars and Planets, above the Atmosphere, from 0.1 to 100.0 Microns. Applied Optics, Vol. 1, No. 4, July 1962.
4. Lockheed Report LMSC/D931340, "The Space Telescope End-to-End Optical Performance Analysis," February 1984.
5. Landauer, F. P., et al.: An 800 x 800 CCD Imager for Space-Borne Scientific Imaging. Proc. 1978 Gov. Micro Applications Conf., Monterey, CA, November 1978.

TABLE 1. PHOTOPIC EYE RESPONSE

λ (10^{-5} cm)	$K(\lambda)$	λ (10^{-5} cm)	$K(\lambda)$	λ (10^{-5} cm)	$K(\lambda)$
3.8	0.00004	5.1	0.503	6.4	0.175
3.9	0.00012	5.2	0.71	6.5	0.107
4.0	0.0004	5.3	0.862	6.6	0.061
4.1	0.0012	5.4	0.954	6.7	0.032
4.2	0.004	5.5	0.995	6.8	0.017
4.3	0.0116	5.6	0.995	6.9	0.0082
4.4	0.023	5.7	0.952	7.0	0.0041
4.5	0.038	5.8	0.87	7.1	0.0021
4.6	0.06	5.9	0.757	7.2	0.00105
4.7	0.09	6.0	0.631	7.3	0.00051
4.8	0.139	6.1	0.503	7.4	0.00025
4.9	0.208	6.2	0.381	7.5	0.00012
5.0	0.323	6.3	0.265	7.6	0.00006

TABLE 2. QUANTUM EFFICIENCIES

λ (10^{-5} cm)	$Q(\lambda)$	λ (10^{-5} cm)	$Q(\lambda)$	λ (10^{-5} cm)	$Q(\lambda)$
1.0	0.0*	4.5	0.08*	8.0	0.12
1.5	0.028*	5.0	0.12	8.5	0.105*
2.0	0.05	5.5	0.15*	9.0	0.09
2.5	0.05	6.0	0.18	9.5	0.08*
3.0	0.05	6.5	0.18*	10.0	0.07
3.5	0.06	7.0	0.18	10.5	0.045*
4.0	0.04	7.5	0.15*	11.0	0.02

*Interpolated Values

TABLE 3. FILTER FUNCTION

λ (10^{-5} cm)	$F(\lambda)$
5.0	0.278*
5.5	0.899
6.0	0.578
6.5	0.0835*

*The values were divided by 2 for using trapezoidal rule in numerical integration over wavelength.




FIGURE 1. RELATIONSHIP OF EXTENDED SOURCE
IMAGE TO CCD CONFIGURATION

APPENDIX A

OUTPUT EXAMPLE

ENTER EXTENDED SOURCE VISUAL MAGNITUDE
?25.
ENTER CHOICE OF SPECTRAL DISTRIBUTION FOR SOURCE
1-BLACKBODY OR 0-FLAT
?0.
ENTER LOWER AND UPPER WAVELENGTHS IN CM
?1.E-5,11.E-5
ENTER NUMBER OF WAVELENGTH CALCULATIONS
?21.
IS THERE A TABLE OF QUANTUM EFFICIENCIES?(YES=1;NO=0)
?1.
IS THERE A TABLE OF FILTER FUNCTIONS?(YES=1;NO=0)
?1.
ENTER TELESCOPE ENTRANCE PUPIL DIAMETER(CM) AND
SYSTEM FOCAL RATIO
?240.,30.
ENTER COSMIC BACKGROUND VISUAL MAGNITUDE
?23.
ENTER X,Y WIDTHS(ARC SECONDS) OF SOURCE
?0.25,0.25
ENTER X,Y WIDTHS(ARC SECONDS) OF A PEL
?0.043,0.043
ENTER X,Y PEL CENTER SEPARATIONS(ARC SECONDS)
?0.043,0.043
ENTER X,Y DETECTOR OFFSET(ARC SECONDS)
X,Y VALUES MUST BE -,>0,BUT < PEL WIDTHS
?0.004,0.004
ENTER FRACTION OF A PEL FILLED FOR (S/N) PER PEL
?1.
ENTER RMS READOUT NOISE(ELECTRONS/PEL)
?18.
ENTER DETECTOR TEMPERATURE(KELVIN)
?178.
ENTER OBSERVATION START AND END TIMES(SECONDS)
?3600.,50400.
ENTER NUMBER(>1) OF TIME CALCULATIONS
?14.
ENTER START AND END SIGNAL TO NOISE RATIOS(S/N)
?1.,10.
ENTER NUMBER(>1) OF (S/N) CALCULATIONS
?10.
CLEAR SCREEN AND RETURN FOR OUTPUT
?

TELESCOPE ENTRANCE PUPIL DIAMETER 240.000 CM
OPTICAL SYSTEM FOCAL RATIO 30.00

SOURCE CHARACTERISTICS

FLAT DISTRIBUTION .64918292E 00 DETECT.ELEC/SEC/ARCSEC**2
EXTENDED SOURCE MAGNITUDE 25.00
ERGS/SEC/CM**2/ARCSEC**2 OUTSIDE ATMOSPHERE .37353315E-10
SOURCE DIMENSIONS(ARC SECONDS):X= .2500 Y= .2500

BACKGROUND CHARACTERISTICS

FLAT DISTRIBUTION .40960512E 01 DETECT.ELEC/SEC/ARCSEC**2
BACKGROUND MAGNITUDE 23.00
ERGS/SEC/CM**2/ARCSEC**2 OUTSIDE ATMOSPHERE .23568268E-14

SYSTEM SPECTRAL CHARACTERISTICS

WAVELENGTH(CM)	QUAN.EFFIC.	FILTER FUNCTION
.49999959E-04	.12000000E 00	.27800000E 00
.54999953E-04	.14999998E 00	.89899999E 00
.59999948E-04	.18000001E 00	.57800001E 00
.64999942E-04	.18000001E 00	.83500028E-01

DETECTOR CHARACTERISTICS

PEL DIMENSIONS(ARC SECONDS):X= .0430 Y= .0430
PEL CENTER SEPARATION(ARC SECONDS):X= .0430 Y= .0430
DETECTOR OFFSET(ARC SECONDS):X= .0040 Y= .0040
ARRAY SIZE EFFECTED BY SOURCE:X= 6 BY Y= 6
RMS READOUT NOISE(ELECTRONS/PEL):18.00
DETECTOR TEMPERATURE(KELVIN):178.0000
DARK CURRENT(ELECTRONS/PEL/SECOND):.53327158E-02

SIGNAL TO NOISE RATIO AS A FUNCTION OF OBSERVATION TIME
PER ARRAY(S/N) AND PER FRACTION OF A PEL(S/N/P), WHERE
THE FRACTION IS 1.0000

TIME(SECONDS)	(S/N)	(S/N/P)
.36000000E 04	.12579422E 01	.22321123E 00
.72000000E 04	.23616362E 01	.41894031E 00
.10800000E 05	.33490124E 01	.59396935E 00
.14400000E 05	.42455339E 01	.75284463E 00
.18000000E 05	.50690994E 01	.89875793E 00
.21600000E 05	.58327589E 01	.10340338E 01
.25200000E 05	.65463123E 01	.11604156E 01
.28800000E 05	.72172985E 01	.12792473E 01
.32400000E 05	.78516531E 01	.13915796E 01
.36000000E 05	.84541206E 01	.14982586E 01
.39600000E 05	.90285597E 01	.15999680E 01
.43200000E 05	.95781527E 01	.16972752E 01
.46800000E 05	.10105552E 02	.17906475E 01
.50400000E 05	.10612990E 02	.18804836E 01

(S/N)	SECONDS(ARRAY)	SECONDS(PIXEL)
1.00	.28196663E 04	.20660961E 05
2.00	.59726367E 04	.55399445E 05
3.00	.94849102E 04	.10702656E 06
4.00	.13381652E 05	.17698156E 06
5.00	.17686836E 05	.26591025E 06
6.00	.22422988E 05	.37410606E 06
7.00	.27610980E 05	.50170906E 06
8.00	.33269922E 05	.64879038E 06
9.00	.39417090E 05	.81538931E 06
10.00	.46068000E 05	.10015279E 07

DO YOU WISH TO CHANGE SOURCE MAGNITUDE?(YES=1;NO=0)

70.

DO YOU WISH TO CHANGE BACKGROUND MAGNITUDE?(YES=1;NO=0)
70.

DO YOU WISH TO CHANGE DETECTOR EFFECTS?(YES=1;NO=0)
70.

DID YOU MAKE ANY CHANGES?(YES=1;NO=0)
70.

STOP 0

APPENDIX B
PROGRAM LISTING

07:29 AUG 10, '84 DC/EXTSRC.JAYROE

```

C      ***** THIS PROGRAM COMPUTES THE SIGNAL TO NOISE RATIO *****
C      ***** OUTSIDE THE EARTH'S ATMOSPHERE FOR A RECTANGULAR *****
C      ***** UNIFORM EXTENDED SOURCE OF VISUAL MAGNITUDE XSM *****
C      ***** EMBEDDED IN A BACKGROUND OF VISUAL MAGNITUDE XMB *****
C      ***** AND FOR A RANGE OF OBSERVATION TIMES. THE CHAR- *****
C      ***** ACTERISTICS OF THE RECTANGULAR DETECTOR ARRAY *****
C      ***** THAT ARE USED INCLUDE QUANTUM EFFICIENCY, PEL *****
C      ***** DIMENSION, DEAD SPACE, RMS READOUT NOISE AND MEAN *****
C      ***** DARK CURRENT. THE SOURCE AND BACKGROUND CAN HAVE *****
C      ***** A FLAT SPECTRAL DISTRIBUTION OR A BLACKBODY DIS- *****
C      ***** TRIBUTION, IN WHICH CASE THE EFFECTIVE TEMPERA- *****
C      ***** TURE OF THE SOURCE AND BACKGROUND MUST BE INPUT. *****
C      ***** THE OBSERVATION TELESCOPE/DETECTOR SYSTEM HAS AN *****
C      ***** ENTRANCE PUPIL DIAMETER DP AND A FOCAL RATIO FR. *****
C      ***** OPTICAL FILTERS CAN ALSO BE USED WITH FILTER *****
C      ***** FUNCTIONS FF. *****
      DIMENSION XK(40), QUEF(50), FF(50)
      WRITE(102,6000)
6000  FORMAT(' ENTER EXTENDED SOURCE VISUAL MAGNITUDE')
      READ(101,5000)XSM
5000  FORMAT(20G)
C      ***** UNITS OF XSM ARE (10**(-5))ERGS/CM*CM/SEC *****
      XIMS=0.37353*10.0**(-0.4*XSM)
      WRITE(102,6004)
6004  FORMAT(/, ' ENTER CHOICE OF SPECTRAL DISTRIBUTION FOR SOURCE')
      WRITE(102,6005)
6005  FORMAT(' 1=BLACKBODY OR 0=FLAT')
      DLA=1.E-6
      HC=1.9865E-16
      HCOK=1.4388
      READ(101,5000)KRESP
      IF(KRESP.EQ.1)GO TO 100
C      ***** PHOTOPIC INTEGRAL OF K(L)DL FROM (3.8-7.6)E-5 *****
C      ***** IS 1.0682E-5 CM. *****
      XIMLS=XIMS/(1.0682*HC)
      GO TO 130
100  SUM=0.0
      DO 110 IK=1,39
110  READ(20,5000)XK(IK)
      WRITE(102,6009)
6009  FORMAT(/, ' ENTER SOURCE EFFECTIVE TEMPERATURE(KELVIN)')
      READ(101,5000)SEFT
C      ***** COMPUTE PHOTOPIC/BLACKBODY INTEGRAL(SIMPSON'S RULE) *****
      X=3.8E-5
      Y=HCOK/(X*SEFT)
      SUM=SUM+XK(1)/((EXP(Y)-1.)*X**5)
      X=X+DLA
      DO 120 IK=2,38
      ICHEK=2*(IK/2)
      FAC=2.

```

```

      IF(IK.EQ.ICHEK)FAC=4.
      Y=HCOK/(X*SEFT)
      SUM=SUM+FAC*XK(IK)/((EXP(Y)-1.)*X**5)
      X=X+DLA
120  CONTINUE
      Y=HCOK/(X*SEFT)
      SUMT=SUM+XK(39)/((EXP(Y)-1.)*X**5)
      XIMLS=XIMS*SUMT*DLA/(3.*HC)
130  WRITE(102,6013)
6013 FORMAT(/,' ENTER LOWER AND UPPER WAVELENGTHS IN CM')
      READ(101,5000)WAVL,WAVH
      WAV=WAVL
      WRITE(102,6017)
6017 FORMAT(/,' ENTER NUMBER OF WAVELENGTH CALCULATIONS')
      READ(101,5000)NINC
      IF(NINC.EQ.1)GO TO 180
      X=NINC-1
      DL=(WAVH-WAVL)/X
      DO 140 I=1,21
      QUEF(I)=1.
      FF(I)=1.
140  CONTINUE
      FF(1)=0.5
      FF(21)=0.5
      WRITE(102,6021)
6021 FORMAT(/,' IS THERE A TABLE OF QUANTUM EFFICIENCIES?(YES=1;NO=0)')
      READ(101,5000)IRESP
      IF(IRESP.EQ.0)GO TO 160
      DO 150 I=1,NINC
150  READ(22,5000)QUEF(I)
160  WRITE(102,6025)
6025 FORMAT(/,' IS THERE A TABLE OF FILTER FUNCTIONS?(YES=1;NO=0)')
      READ(101,5000)IRESP
      IF(IRESP.EQ.0)GO TO 200
      DO 170 I=1,NINC
170  READ(24,5000)FF(I)
      GO TO 200
180  WRITE(102,6029)
6029 FORMAT(/,' ENTER MONOCHROMATIC QUANTUM EFFICIENCY')
      READ(101,5000)QUEF(1)
      WRITE(102,6033)
6033 FORMAT(/,' ENTER MONOCHROMATIC FILTER FUNCTION')
      READ(101,5000)FF(1)
      IF(KRESP.EQ.1)GO TO 190
      SUM=XIMLS*WAVL*QUEF(1)*FF(1)
      GO TO 260
190  Y=HCOK/(WAVL*SEFT)
      SUM=XIMLS*QUEF(1)*FF(1)/((EXP(Y)-1.)*WAVL**4)
      GO TO 260
200  MINC=NINC-1
      IF(KRESP.EQ.1)GO TO 230
      SUM=WAV*QUEF(1)*FF(1)

```

```

      IF(MINC.EQ.2)GO TO 220
      DO 210 I=2,MINC
      WAV=WAV+DL
      PROD=QUEF(I)*FF(I)
      IF(PROD.LT.1.E-4)GO TO 210
      SUM=SUM+WAV*PROD
210  CONTINUE
220  WAV=WAV+DL
      SUM=SUM+WAV*QUEF(NINC)*FF(NINC)
      SUM=SUM*DL*XIMLS
      GO TO 260
230  Y=HCOK/(WAV*SEFT)
      SUM=QUEF(1)*FF(1)/((EXP(Y)-1.)*WAV**4)
      IF(MINC.EQ.2)GO TO 250
      DO 240 I=2,MINC
      WAV=WAV+DL
      PROD=QUEF(I)*FF(I)
      IF(PROD.LT.1.E-4)GO TO 240
      Y=HCOK/(WAV*SEFT)
      SUM=SUM+PROD/((EXP(Y)-1.)*WAV**4)
240  CONTINUE
250  WAV=WAV+DL
      Y=HCOK/(WAV*SEFT)
      SUM=SUM+QUEF(NINC)*FF(NINC)/((EXP(Y)-1.)*WAV**4)
      SUM=SUM*DL*XIMLS
260  WRITE(102,6037)
6037  FORMAT(/,'ENTER TELESCOPE ENTRANCE PUPIL DIAMETER(CM) AND')
      WRITE(102,6038)
6038  FORMAT('SYSTEM FOCAL RATIO')
      READ(101,5000)DP,FR
      SIGNL=0.7854*DP*DP*SUM
      WRITE(102,6041)
6041  FORMAT(/,'ENTER COSMIC BACKGROUND VISUAL MAGNITUDE')
      READ(101,5000)XMB
      XIMB=0.37353*10.0**(-0.4*XMB-5.)
      WAV=WAVL
      IF(KRESP.EQ.1)GO TO 270
      XIMLB=(XIMB*10.0**5)/(1.0682*HC)
      GO TO 285
270  WRITE(102,6049)
6049  FORMAT(/,'ENTER BACKGROUND EFFECTIVE TEMPERATURE(KELVIN)')
      READ(101,5000)BEFT
      X=3.8E-5
      Y=HCOK/(X*BEFT)
      SUM=XK(1)/((EXP(Y)-1.)*X**5)
      X=X+DLA
      DO 280 IK=2,38
      ICHEK=2*(IK/2)
      FAC=2.
      IF(IK.EQ.ICHEK)FAC=4.
      Y=HCOK/(X*BEFT)
      SUM=SUM+FAC*XK(IK)/((EXP(Y)-1.)*X**5)

```

```

      X=X+DLA
280  CONTINUE
      Y=HCOK/(X*BEFT)
      SUM=SUM+XK(39)/((EXP(Y)-1.)*X**5)
      XIMLB=XIMB*3./(SUM*DLA*HC)
285  IF(NINC.GT.1)GO TO 300
      IF(KRESP.EQ.1)GO TO 290
      SUM=XIMLB*WAV*QUEF(1)*FF(1)
      GO TO 360
290  Y=HCOK/(WAV*BEFT)
      SUM=XIMLB*QUEF(1)*FF(1)/((EXP(Y)-1.)*WAV**4)
      GO TO 360
300  IF(KRESP.EQ.1)GO TO 330
      SUM=WAV*QUEF(1)*FF(1)
      IF(MINC.EQ.2)GO TO 320
      DO 310 I=2,MINC
      WAV=WAV+DL
      PROD=QUEF(I)*FF(I)
      IF(PROD.LT.1.E-4)GO TO 310
      SUM=SUM+WAV*PROD
310  CONTINUE
320  WAV=WAV+DL
      SUM=SUM+WAV*QUEF(NINC)*FF(NINC)
      SUM=SUM*DL*XIMLB
      GO TO 360
330  Y=HCOK/(WAV*BEFT)
      SUM=QUEF(1)*FF(1)/((EXP(Y)-1.)*WAV**4)
      IF(MINC.EQ.2)GO TO 350
      DO 340 I=2,MINC
      WAV=WAV+DL
      PROD=QUEF(I)*FF(I)
      IF(PROD.LT.1.E-4)GO TO 340
      Y=HCOK/(WAV*BEFT)
      SUM=SUM+PROD/((EXP(Y)-1.)*WAV**4)
340  CONTINUE
350  WAV=WAV+DL
      Y=HCOK/(WAV*BEFT)
      SUM=SUM+QUEF(NINC)*FF(NINC)/((EXP(Y)-1.)*WAV**4)
      SUM=SUM*DL*XIMLB
360  BGRND=0.7854*DP*DP*SUM
361  WRITE(102,6053)
6053  FORMAT(/,'ENTER X,Y WIDTHS(ARC SECONDS) OF SOURCE')
      READ(101,5000)XSW,YSW
      WRITE(102,6057)
6057  FORMAT(/,'ENTER X,Y WIDTHS(ARC SECONDS) OF A PEL')
      READ(101,5000)XPW,YPW
      WRITE(102,6061)
6061  FORMAT(/,'ENTER X,Y PEL CENTER SEPARATIONS(ARC SECONDS)')
      READ(101,5000)XPD,YPD
      WRITE(102,6065)
6065  FORMAT(/,'ENTER X,Y DETECTOR OFFSET(ARC SECONDS)')
      WRITE(102,6069)

```

```

6069 FORMAT('X,Y VALUES MUST BE =,>0,BUT < PEL WIDTHS')
      READ(101,5000)XOF,YOF
      XN=(XOF+XSW)/XPD
      NX=XN
      MX=NX
      XNX=NX
      IF(XN.GT.XNX)MX=NX+1
      XM=MX
      PRTX=0.
      IF(XPD.LT.XPW)GO TO 365
      PRTX=(XOF+XSW-XNX*XPD-XPW)/(XPD-XPW)
      IF(PRTX.LE.0)PRTX=0.
365  YN=(YOF+YSW)/YPD
      NY=YN
      MY=NY
      YNY=NY
      IF(YN.GT.YNY)MY=NY+1
      YM=MY
      PRTY=0.
      IF(YPD.LE.YPW)GO TO 367
      PRTY=(YOF+YSW-YNY*YPD-YPW)/(YPD-YPW)
      IF(PRTY.LE.0.)PRTY=0.
367  AS=XSW*YSW+(XPD-XPW)*(YPD-YPW)*(XNX*YNY+PRTX*YNY+PRTY*XNX
      &+PRTX*PRTY)
      AS=AS-YSW*(XPD-XPW)*(XNX+PRTX)-XSW*(YPD-YPW)*(YNY+PRTY)
      AB=XPW*YPW*XM*YM
      TOTS=SIGNL*AS
      TOTB=BGRND*AB
      WRITE(102,6073)
6073 FORMAT(/,'ENTER FRACTION OF A PEL FILLED FOR (S/N) PER PEL')
      READ(101,5000)FRAC
      PELS=SIGNL*FRAC*XPW*YPW
      PELB=BGRND*XPW*YPW
      WRITE(102,6077)
6077 FORMAT(/,'ENTER RMS READOUT NOISE(ELECTRONS/PEL)')
      READ(101,5000)RON
      RN=RON*RON
      WRITE(102,6081)
6081 FORMAT(/,'ENTER DETECTOR TEMPERATURE(KELVIN)')
      READ(101,5000)DTEM
      IF(DTEM.LT.4.)GO TO 370
      FAC=-6705.487/DTEM+4.0737*DTEM/(1108.+DTEM)
      FAC=EXP(FAC)*(29.3E+9)
      DC=FAC*DTEM**1.5
      GO TO 380
370  WRITE(102,6085)
6085 FORMAT(/,'ENTER MEAN DARK CURRENT(ELECTRONS/PEL/SECOND)')
      READ(101,5000)DC
380  WRITE(102,6089)
6089 FORMAT(/,'ENTER OBSERVATION START AND END TIMES(SECONDS)')
      READ(101,5000)STRT,ENDT
      WRITE(102,6093)

```

```

6093 FORMAT(/,'ENTER NUMBER(>1) OF TIME CALCULATIONS')
      READ(101,5000)NT
      X=NT-1
      DT=(ENDT-STRT)/X
      WRITE(102,6094)
6094 FORMAT(/,'ENTER START AND END SIGNAL TO NOISE RATIOS(S/N)')
      READ(101,5000)SNST,SNND
      WRITE(102,6095)
6095 FORMAT(/,'ENTER NUMBER(>1) OF (S/N) CALCULATIONS')
      READ(101,5000)NS
      Y=NS-1
      DN=(SNND-SNST)/Y
      WRITE(102,6096)
6096 FORMAT(/,'CLEAR SCREEN AND RETURN FOR OUTPUT')
      READ(101,5000)DUMMY
      445 WRITE(102,6097)DP
6097 FORMAT(/,'TELESCOPE ENTRANCE PUPIL DIAMETER ',F7.3,' CM')
      WRITE(102,6101)FR
6101 FORMAT('OPTICAL SYSTEM FOCAL RATIO ',F5.2)
      WRITE(102,6105)
6105 FORMAT(/,'SOURCE CHARACTERISTICS')
      IF(KRESP.NE.1)GO TO 390
      WRITE(102,6109)SIGNL
6109 FORMAT(1X,'PLANC DISTRIBUTION',E15.8,' DETECT.ELEC/SEC/ARCSEC**2')
      WRITE(102,6113)SEFT
6113 FORMAT(5X,'EFFECTIVE TEMPERATURE ',F9.3,' DEGREES KELVIN')
      GO TO 400
      390 WRITE(102,6117)SIGNL
6117 FORMAT(1X,'FLAT DISTRIBUTION',E15.8,' DETECT.ELEC/SEC/ARCSEC**2')
      400 WRITE(102,6121)XSM
6121 FORMAT(5X,'EXTENDED SOURCE MAGNITUDE ',F5.2)
      WRITE(102,6122)XIMS
6122 FORMAT(5X,'ERGS/SEC/CM**2/ARCSEC**2 OUTSIDE ATMOSPHERE ',E15.8)
      WRITE(102,6125)XSW,YSW
6125 FORMAT(5X,'SOURCE DIMENSIONS(ARC SECONDS):X=',F8.4,' Y=',F8.4)
      WRITE(102,6129)
6129 FORMAT(/,'BACKGROUND CHARACTERISTICS')
      IF(KRESP.NE.1)GO TO 410
      WRITE(102,6109)BGRND
      WRITE(102,6113)BEFT
      GO TO 420
      410 WRITE(102,6117)BGRND
      420 WRITE(102,6145)XMB
6145 FORMAT(5X,'BACKGROUND MAGNITUDE ',F5.2)
      WRITE(102,6122)XIMB
      WRITE(102,6149)
6149 FORMAT(/,'SYSTEM SPECTRAL CHARACTERISTICS')
      WRITE(102,6153)
6153 FORMAT(4X,'WAVELENGTH(CM)',5X,'QUAN.EFFIC.',3X,'FILTER FUNCTION')
      WAV=WAVL-DL
      DO 430 I=1,NINC
      WAV=WAV+DL

```

```

        PROD=QUEF(I)*FF(I)
        IF(PROD.LT.1.E-4)GO TO 430
        WRITE(102,6157)WAV,QUEF(I),FF(I)
6157  FORMAT(3(2X,E15.8))
        430  CONTINUE
        WRITE(102,6161)
6161  FORMAT(/,'DETECTOR CHARACTERISTICS')
        WRITE(102,6165)XPW,YPW
6165  FORMAT(5X,'PEL DIMENSIONS(ARC SECONDS):X=',F8.4,' Y=',F8.4)
        WRITE(102,6169)XPD,YPD
6169  FORMAT(5X,'PEL CENTER SEPARATION(ARC SECONDS):X=',F8.4,' Y=',F8.4)
        WRITE(102,6173)XOF,YOF
6173  FORMAT(5X,'DETECTOR OFFSET(ARC SECONDS):X=',F8.4,' Y=',F8.4)
        WRITE(102,6177)MX,MY
6177  FORMAT(5X,'ARRAY SIZE EFFECTED BY SOURCE:X=',I2,' BY Y=',I2)
        WRITE(102,6181)RON
6181  FORMAT(5X,'RMS READOUT NOISE(ELECTRONS/PEL):',F5.2)
        IF(DTEM.LT.4.)GO TO 440
        WRITE(102,6185)DTEM
6185  FORMAT(5X,'DETECTOR TEMPERATURE(KELVIN):',F8.4)
        440  WRITE(102,6189)DC
6189  FORMAT(5X,'DARK CURRENT(ELECTRONS/PEL/SECOND):',E15.8)
        WRITE(102,6193)
6193  FORMAT(/,'SIGNAL TO NOISE RATIO AS A FUNCTION OF OBSERVATION',
&' TIME')
        WRITE(102,6197)
6197  FORMAT('PER ARRAY(S/N) AND PER FRACTION OF A PEL(S/N/P),WHERE')
        WRITE(102,6201)FRAC
6201  FORMAT('THE FRACTION IS ',F7.4)
        WRITE(102,6205)
6205  FORMAT(4X,'TIME(SECONDS) ',2X,'      (S/N)      ',2X,'      (S/N/P)')
        PRON=XM*YM*RN
        PDC=XM*YM*DC
        T=STRT-DT
        DO 450 I=1,NT
        T=T+DT
        XNUM=TOTS*T
        XDNUM=XNUM+TOTB*T+PRON+PDC*T
        SNT=XNUM/SQRT(XDNUM)
        XNUM=PELS*T
        XDNUM=XNUM+PELB*T+RN+DC*T
        SNP=XNUM/SQRT(XDNUM)
        WRITE(102,6157)T,SNT,SNP
        450  CONTINUE
        WRITE(102,6206)
6206  FORMAT(/,5X,'(S/N)',4X,'SECONDS(ARRAY)',3X,'SECONDS(PIXEL)')
        SN=SNST-DN
        DO 455 I=1,NS
        SN=SN+DN
        A=SN*SN/(TOTS*TOTS)
        B=A*(TOTS+TOTB+PDC)*0.5
        C=A*PRON

```



```

T1=B+SQRT(B*B+C)
A=SN*SN/(PELS*PELS)
B=A*(PELS+PELB+DC)*0.5
C=A*RN
T2=B+SQRT(B*B+C)
WRITE(102,6207)SN,T1,T2
6207 FORMAT(5X,F5.2,2(2X,E15.8))
455 CONTINUE
WRITE(102,6209)
6209 FORMAT(/,'DO YOU WISH TO CHANGE SOURCE MAGNITUDE?(YES=1;NO=0)')
READ(101,5000)JRESP
IF(JRESP.EQ.0)GO TO 460
WRITE(102,6213)
6213 FORMAT(/,'ENTER NEW SOURCE MAGNITUDE')
READ(101,5000)XSM
XNEW=0.37353*10.0**(-0.4*XSM)
XNEW=XNEW/XIMS
SIGNL=SIGNL*XNEW
TOTS=TOTS*XNEW
PELS=PELS*XNEW
460 WRITE(102,6217)
6217 FORMAT(/,'DO YOU WISH TO CHANGE BACKGROUND MAGNITUDE?(YES=1;',
&'NO=0)')
READ(101,5000)JRESP
IF(JRESP.EQ.0)GO TO 470
WRITE(102,6221)
6221 FORMAT(/,'ENTER NEW BACKGROUND MAGNITUDE')
READ(101,5000)XMB
XNEW=0.37353*10.0**(-0.4*XMB)
XNEW=XNEW/XIMB
BGRND=BGRND*XNEW
TOTB=TOTB*XNEW
PELB=PELB*XNEW
470 WRITE(102,6225)
6225 FORMAT(/,'DO YOU WISH TO CHANGE DETECTOR EFFECTS?(YES=1;NO=0)')
READ(101,5000)JRESP
IF(JRESP.EQ.1)GO TO 361
WRITE(102,6229)
6229 FORMAT(/,'DID YOU MAKE ANY CHANGES?(YES=1;NO=0)')
READ(101,5000)JRESP
IF(JRESP.EQ.1)GO TO 445
STOP
END

```

1. REPORT NO. NASA TP-2396		2. GOVERNMENT ACCESSION NO.		3. RECIPIENT'S CATALOG NO.	
4. TITLE AND SUBTITLE Algorithm for Astronomical, Extended Source, Signal-to-Noise Ratio Calculations				5. REPORT DATE November 1984	
				6. PERFORMING ORGANIZATION CODE	
7. AUTHOR(S) R. R. Jayroe				8. PERFORMING ORGANIZATION REPORT #	
9. PERFORMING ORGANIZATION NAME AND ADDRESS George C. Marshall Space Flight Center Marshall Space Flight Center, Alabama 35812				10. WORK UNIT NO. M-468	
				11. CONTRACT OR GRANT NO.	
				13. TYPE OF REPORT & PERIOD COVERED Technical Paper	
12. SPONSORING AGENCY NAME AND ADDRESS National Aeronautics and Space Administration Washington, D.C. 20546				14. SPONSORING AGENCY CODE	
15. SUPPLEMENTARY NOTES Prepared by Information and Electronic System Laboratory, Science and Engineering Directorate					
16. ABSTRACT An algorithm was developed to simulate the expected signal-to-noise ratio as a function of observation time in the charge coupled device detector plane of an optical telescope located outside the Earth's atmosphere for an extended, uniform astronomical source embedded in a uniform cosmic background. By choosing the appropriate input values, the expected extended source signal-to-noise ratios can be computed for the Hubble Space Telescope using the Wide Field/Planetary Camera science instrument.					
17. KEY WORDS Astronomical Signal-to-Noise Ratio Hubble Space Telescope Wide Field/Planetary Camera Computer Algorithm			18. DISTRIBUTION STATEMENT Unclassified — Unlimited STAR Category: 61, 74, 39		
19. SECURITY CLASSIF. (of this report) Unclassified		20. SECURITY CLASSIF. (of this page) Unclassified		21. NO. OF PAGES 22	
				22. PRICE A02	

National Aeronautics and
Space Administration

Washington, D.C.
20546

Official Business

Penalty for Private Use, \$300

THIRD-CLASS BULK RATE

Postage and Fees Paid
National Aeronautics and
Space Administration
NASA-451



NASA

POSTMASTER: If Undeliverable (Section 158
Postal Manual) Do Not Return
